# $Id: PKGBUILD 81142 2012-12-12 01:39:15Z eric $
# Maintainer: Sven-Hendrik Haase <sh@lutzhaase.com>

pkgname=megaglest-data
_pkgbasename=megaglest
pkgver=3.7.1
pkgrel=1
pkgdesc="Fork of Glest, a 3D real-time strategy game in a fantastic world (data files)"
arch=('any')
url="http://sourceforge.net/projects/megaglest/"
license=('custom:CCBYSA3.0')
makedepends=('cmake' 'xz')
source=("http://downloads.sourceforge.net/${_pkgbasename}/${_pkgbasename}-data-${pkgver}.tar.xz")
md5sums=('a73205fe02b684547dfe1fedc2d712b2')

build() {
  cd "${srcdir}"/"${_pkgbasename}"-"${pkgver}"

  [[ -d build ]] && rm -r build
  mkdir build && cd build
  cmake -DCMAKE_INSTALL_PREFIX=/usr ..
}

package() {
  cd "${srcdir}"/"${_pkgbasename}"-"${pkgver}"

  cd build
  make DESTDIR="${pkgdir}" install

  install -Dm644 ../docs/README.data-license.txt "${pkgdir}"/usr/share/licenses/${pkgname}/LICENSE
}

# vim: sw=2:ts=2 et:
